

# Matrici

Dal punto di vista algebrico i vettori, come i punti, sono terne ordinate di numeri reali:  $(x_1, x_2, x_3)$ . Questo almeno nello spazio, che ha tre dimensioni. Sul piano, provvisto di sole due dimensioni, sono invece coppie ordinate di numeri:  $(x_1, x_2)$ .

Nulla vieta di considerare quaterne o cinque ordinate di numeri reali, nel qual caso si passa agli iperspazi. Uno spazio di  $n$  dimensioni avrà i punti rappresentati da vettori con  $n$  elementi:  $(x_1, x_2, \dots, x_i, \dots, x_n)$ .

Tuttavia l'algebra vettoriale fa pure uso di ulteriori insiemi, dove, in luogo di un solo indice, si utilizzano due indici. Si tratta delle matrici, che possono intendersi ciascuna come un insieme di  $m$  vettori a  $n$  dimensioni ciascuno. Esse sono infatti ordinate per righe e per colonne, ciascuna colonna rappresenta un vettore.

$$\begin{array}{cccc} x_{11} & x_{21} & \cdots & x_{m1} \\ x_{12} & x_{22} & \cdots & x_{m2} \\ \cdots & \cdots & \cdots & \cdots \\ x_{1n} & x_{2n} & \cdots & x_{mn} \end{array}$$

I valori parametrici della finestra di dialogo degli oggetti gsm, ricavati con lo script GDL, possono essere organizzati come matrici. Basta cliccare sulla settima casella del parametro, che rappresenta appunto in modo schematico una matrice. I valori assegnabili non rappresenteranno allora più un unico parametro, ma piuttosto un insieme di  $m \cdot n$  parametri, ciascuno definito dai suoi indici  $i, j$ . A ciascuno di questi parametri può essere attribuito soltanto un unico valore, entro la casella matrice relativa al parametro indicizzato.

Si parte da un parametro qualsiasi nella finestra di dialogo, cui abbiamo attribuito un nome, ad esempio var. Poi si clicca sull'elemento matrice, si apre allora la finestra di dialogo dove dobbiamo fare due cose:

1: Definire le dimensioni  $m \cdot n$  della matrice, cioè il numero delle colonne e il numero delle righe. Questa operazione stabilisce l'effettivo numero di variabili richiamabili dal valore parametrico denominato var.

2: Attribuire il valore ai singoli elementi (saranno appunto  $m \cdot n$ ) della matrice.

La variabile cui abbiamo dato il nome var, viene poi interpretata come se avessimo scritto nel **Testo Master** :

`DIM var[m][n]`

dove qui  $m, n$  vanno intesi come effettivi numeri (non variabili), mentre il comando

`var[1][2]`

viene interpretato come il valore che abbiamo attribuito alla casella colonna 1, riga 2 nella finestra di dialogo relativamente al parametro var, salvo che non gli attribuiamo un altro valore nello script.

Il comando DIM, cui va fatto seguire, dopo uno spazio, l'elenco di lettere che definiscono la variabile più una o due parentesi quadre, comunica al calcolatore che quella specifica variabile locale è da intendersi come variabile del tipo matrice, a uno o due indici. Se al nome scelto per la variabile si fa seguire una sola coppia di parentesi quadre avremo una variabile a un solo indice, se seguono due coppie di parentesi, avremo una autentica matrice a due indici. Se entro le parentesi non viene specificato alcun valore: `DIM var[][]`, la variabile potrà assumere tutti i valori definiti successivamente entro lo script (matrice dinamica). I valori non definiti valgono zero. Se vengono invece prefissati numeri interi entro le parentesi, definendone le dimensioni, verranno accettati soltanto quei valori con indici compresi entro la dimensione della matrice.

Ritornando alla finestra di dialogo, gli  $m \times n$  valori parametrici, compresi entro la matrice chiamata var, potevano anche essere scritti nella finestra di dialogo, separatamente, come  $m \times n$  singoli parametri, ciascuno denominato in modo differente:

```
var11  var21  ...  varm1  (riga 1 della matrice var originaria)
...
Var1n  var2n  ...  varmn  (riga n)
```

Analogamente nel testo si potrebbero elencare direttamente le singole variabili, ciascuna con un nome specifico, invece di definire un parametro del tipo matrice, quale appunto

```
DIM var[][]
```

che poi necessita comunque di specificare i singoli valori

```
var[1][1]=...
var[1][2]=...
...
```


Tuttavia, in alcuni casi, l'utilizzo della funzione DIM è indispensabile, ad esempio per calcolare sommatorie.

Come esempio utilizziamo le variabili di tipo matrice per verificare la potenza di un numero complesso:  $(a + ib)^n$ . Questa potenza esprime una rotazione del vettore (a,b) pari a n volte il valore angolare iniziale, mentre il modulo finale si ricava elevando alla potenza di n il modulo iniziale.

Utilizzando il comando di tipo matrice potremo scrivere:<sup>1</sup>

```
DIM c[]
DIM d[]
c[1]=a
d[1]=b
line2 0,0,c[1],d[1]
for i=2 to n
c[i]=a*c[i-1]-b*d[i-1]
d[i]=b*c[i-1]+a*d[i-1]
line2 0,0,c[i],d[i]
next i

for i=1 to n-1
line2 c[i],d[i],c[i+1],d[i+1]
next i
```



In alternativa, i raggi vettoriali che esprimono le potenze successive, possono essere descritti nel modo seguente:

```
an=atn(b/a)      !angolo del vettore rispetto x+
mab=sqr(a^2+ b^2) !modulo del vettore rispetto x+
for i=1 to n-1
line2 mab^i*cos(an*i),      mab^i*sin(an*i),
      mab^(i+ 1)*cos(an*(i+ 1)),mab^(i+ 1)*sin(an*(i+ 1))
next i
```

<sup>1</sup> Il comando `line2 x1,y1,x2,y2` traccia in 2D un segmento che collega i punti (x1,y1) e (x2,y2).